

Example formulas

Twenty-three working formulas from the FarmQA support team: phosphorus, potassium, lime, and nitrogen recommendations you can copy into your own account and adjust. Each one lists its settings, its variables, and an expression you can copy and paste.

Use an example

1. Select **Fields**, then select the **Formulas** tab, then select **New formula**.
2. Set the **Output type** and output details listed for the example.
3. Add every variable in the example's table, using the same names. The expression refers to variables by name, so a name that doesn't match causes an error.
4. Copy the expression from this document and paste it into **Expression**.
5. Change the numeric values to match your own recommendation guidelines, then select **Save**.

Good to know

- Formulas require the Advanced Prescriptions add-on.
- Expressions use spreadsheet-style functions such as IF, AND, OR, and MAX. The leading = is optional, and capitalization doesn't matter.
- Variable names can use letters, numbers, and underscores, must start with a letter or underscore, and can't match a function name like IF.
- A formula that calculates a nutrient, product, or seeding rate never returns a negative rate. Anything below zero comes out as 0.
- These examples reflect the recommendation guidelines named in each description. Check the values against the guidelines you follow before you run them on grower fields.

What's inside

Phosphorus: build, maintain, and draw down	7 crops
Potassium: CEC-adjusted build, maintain, and draw down	7 crops
Lime	3 formulas
Sugar beet nitrogen	2 formulas
Tri-State recommendations	4 formulas

Formula names match how they appear in the FarmQA support account, so they line up with the screenshots.

Phosphorus: build, maintain, and draw down

Seven crop formulas share one expression and differ only in their numeric values. Below the critical level, the rate builds the soil test up and replaces what the crop removes. In the maintenance range, it replaces crop removal only. Above that, it tapers toward zero, and it returns 0 once the soil test is 25 ppm above the critical level.

Output type	Nutrient rate
Nutrient output	Phosphorus, Pounds per Acre
Description	MSU Nutrient Rec

Variables

Variable	Type	Setting	What it does
CL	Numeric	Crop value (table below)	Critical soil test level, in ppm. Below it, the rate adds build-up of 5 lb per ppm short.
YP	Numeric, Prompt for value on	Crop value (table below)	Yield goal. You enter it each time you run the formula.
CR	Numeric	Crop value (table below)	Phosphorus removed per unit of yield. $YP \times CR$ is the maintenance rate.
PL	Numeric	Crop value (table below)	Width of the maintenance range above CL, in ppm.
DDL	Numeric	Crop value (table below)	Controls how quickly the rate tapers above the maintenance range.
P1	Soil nutrient	Phosphorus, Parts Per Million (ppm)	The soil test phosphorus value.

Values by crop

Formula	CL	YP	CR	DDL	PL
Alfalfa Phosphate	25	5	13	10	15
Corn Grain Phosphate	15	175	0.37	10	15
Corn Silage Phosphate	15	175	3.3	10	15
Dry Beans Phosphate	15	30	1.2	30	25
Soybeans Phosphate	15	50	0.8	10	15
Sugar Beets Phosphate	15	35	1.3	10	15
Wheat Phosphate	25	100	0.63	10	15

Sugar Beets Phosphate names its soil variable ST instead of P1. To use that name, replace P1 with ST everywhere in the expression.

Expression

```
=IF(P1<CL,
  ((CL-P1)*5)+(YP*CR),
  IF(AND(CL<=P1, P1<=CL+PL),
    YP*CR,
    IF(AND(P1>CL+PL, P1<=CL+25),
      (YP*CR) - ((YP*CR)*(P1-(CL+PL))/DDL),
      0
    )
  )
)
```

The screenshot shows the 'Wheat Phosphate' formula editor. The interface includes fields for Name, Description, Output type, Nutrient output, and various variable categories (Numeric, Soil Nutrient, Crop Nutrient Coefficient, Crop, and Assigned Growing Season Crop Variables). The formula is displayed in the Expression field at the bottom.

Name: Wheat Phosphate
Description: MSU Nutrient Rec

Output type: Nutrient rate
Nutrient output: Phosphorus P Pounds per Acre ☐ Expression outputs a value in a different unit
☐ Use interpolated nutrient layers

NUMERIC VARIABLES

Variable	Value	Zone-based	Prompt for value
CL	25	☐	☐
YP	100	☐	☒
CR	0.63	☐	☐
DDL	10	☐	☐
PL	15	☐	☐

SOIL NUTRIENT VARIABLES

P1 Phosphorus P Parts Per Million (ppm) All All

CROP NUTRIENT COEFFICIENT VARIABLES

CROP VARIABLES

ASSIGNED GROWING SEASON CROP VARIABLES

Expression:

```
=IF(P1<CL,
  ((CL-P1)*5)+(YP*CR),
  IF(AND(CL<=P1, P1<=CL+PL),
    YP*CR,
    IF(AND(P1>CL+PL, P1<=CL+25),
      (YP*CR) - ((YP*CR)*(P1-(CL+PL))/DDL),
      0
    )
  )
)
```

Wheat Phosphate in the formula editor

Potassium: CEC-adjusted build, maintain, and draw down

Seven crop formulas share one expression. The critical level adjusts to the soil's cation exchange capacity: $75 + (2.5 \times \text{CEC})$ ppm. Below it, the rate builds the soil test up and replaces crop removal. In the maintenance range, it replaces crop removal only. Above that, it tapers toward zero, and it returns 0 once the soil test is 50 ppm above the critical level.

Output type	Nutrient rate
Nutrient output	Potassium, Pounds per Acre
Description	MSU Nutrient Rec

Variables

Variable	Type	Setting	What it does
YP	Numeric, Prompt for value on	Crop value (table below)	Yield goal. You enter it each time you run the formula.
CR	Numeric	Crop value (table below)	Potassium removed per unit of yield. $\text{YP} \times \text{CR}$ is the maintenance rate.
PL	Numeric	Crop value (table below)	Width of the maintenance range above the critical level, in ppm.
DDL	Numeric	Crop value (table below)	Controls how quickly the rate tapers above the maintenance range.
ST	Soil nutrient	Potassium, Parts Per Million (ppm)	The soil test potassium value.
CEC	Soil nutrient	Cation Exchange Capacity, mEq Per 100 Gram (mEq/100g)	Sets the critical level and scales the build-up rate.

Values by crop

Formula	YP	CR	DDL	PL	Prompts for
Alfalfa Potash	5	50	50	0	YP
Corn Grain Potash	175	0.27	20	30	YP, CR, DDL, PL
Corn Silage Potash	175	8	20	0	YP
Dry Beans Potash	30	1.6	20	30	YP
Soybeans Potash	50	1.4	20	30	YP
Sugar Beets Potash Rec	35	3.3	20	30	YP
Wheat Potash	100	0.37	20	30	YP

Expression

```
=IF(ST<(75+(2.5*CEC)),
  ((75+(2.5*CEC))-ST)*(1+0.05*CEC)+(YP*CR),
  IF(AND(ST>=(75+(2.5*CEC)),ST<=((75+(2.5*CEC))+PL)),
    YP*CR,
    IF(AND(ST>((75+(2.5*CEC))+PL),ST<((75+(2.5*CEC))+50)),
      (YP*CR)-(((YP*CR)*(ST-((75+(2.5*CEC))+PL)))/DDL),
      0
    )
  )
)
```

The screenshot shows the 'Wheat Potash' formula editor. The 'Name' field is 'Wheat Potash' and the 'Description' is 'MSU Nutrient Rec'. The 'Output type' is 'Nutrient rate'. The 'Nutrient output' is 'Potassium' (K) with units 'Pounds per Acre'. There is a checkbox for 'Expression outputs a value in a different unit' which is unchecked, and a checkbox for 'Use interpolated nutrient layers' which is also unchecked.

NUMERIC VARIABLES

- YP**: Value 100, Zone-based checkbox unchecked, Prompt for value checkbox checked.
- CR**: Value 0.37, Zone-based checkbox unchecked, Prompt for value checkbox unchecked.
- DDL**: Value 20, Zone-based checkbox unchecked, Prompt for value checkbox unchecked.
- PL**: Value 30, Zone-based checkbox unchecked, Prompt for value checkbox unchecked.

SOIL NUTRIENT VARIABLES

- ST**: Nutrient Potassium (K), Unit Parts Per Million (ppm), All, All, Value field empty.
- CEC**: Nutrient Cation Exchange Capacity (CEC), Unit mEq Per 100 Gram (mEq/100g), All, All, Value field empty.

CROP NUTRIENT COEFFICIENT VARIABLES

CROP VARIABLES

ASSIGNED GROWING SEASON CROP VARIABLES

Expression

```
=IF(ST<(75+(2.5*CEC)),
  ((75+(2.5*CEC))-ST)*(1+0.05*CEC)+(YP*CR),
  IF(AND(ST>=(75+(2.5*CEC)),ST<=((75+(2.5*CEC))+PL)),
    YP*CR,
    IF(AND(ST>((75+(2.5*CEC))+PL),ST<((75+(2.5*CEC))+50)),
      (YP*CR)-(((YP*CR)*(ST-((75+(2.5*CEC))+PL)))/DDL),
      0
    )
  )
)
```

Wheat Potash in the formula editor

Lime

All Crop Lime Rec

Returns a lime rate in steps based on soil pH, from 6,000 at pH 5.5 or below down to 0 above pH 6.8.

Output type	Product rate
Product	Lime
Description	MSU Nutrient Rec

Variable	Type	Setting	What it does
pH	Soil nutrient	pH, pH Value	The soil test pH.

Expression

```
=IF(pH<=5.5,6000,
  IF(pH<=5.8,5000,
    IF(pH<=6.0,4000,
      IF(pH<=6.3,3000,
        IF(pH<=6.8,2000,
          IF(pH>6.9,0,0)
        )
      )
    )
  )
)
```

New formula

Duplicate

Name

All Crop Lime Rec

Description

MSU Nutrient Rec

Output type

Product rate

Product

Lime

None

☐ Use interpolated nutrient layers ⓘ

NUMERIC VARIABLES ⓘ

+ Add

SOIL NUTRIENT VARIABLES ⓘ

pH

pH

pH Value

All

All

+ Add

CROP NUTRIENT COEFFICIENT VARIABLES ⓘ

+ Add

CROP VARIABLES ⓘ

+ Add Generate

ASSIGNED GROWING SEASON CROP VARIABLES ⓘ

+ Add

Expression

```

~IF (pH<=5.5,6000,
  IF (pH<=5.8,5000,
    IF (pH<=6.0,4000,
      IF (pH<=6.3,3000,
        IF (pH<=6.8,2000,
          IF (pH>6.9,0,0)
        )
      )
    )
  )
)

```

All Crop Lime Rec in the formula editor

FarmQA · Example formulas

7 of 16

Lime Rec - MSU

Calculates tons of lime per acre from the lime index and the pH you're targeting (6.0, 6.5, or 6.8). If no lime index is entered, it falls back to a pH-only rate: 1 ton when pH is 6.0 to 6.2, 2 tons below 6.0, and 0 above 6.2.

Output type	Product rate
Product	Lime, tons / ac
Description	pH Targets: 6.0, 6.5, 6.8

Variable	Type	Setting	What it does
LI	Numeric, Prompt for value on	Default 69	Lime index. Enter it as a number when you run the formula.
target_pH	Numeric, Prompt for value on	Default 6.5	The pH you're liming to: 6.0, 6.5, or 6.8.
pH	Soil nutrient	pH, pH Value	The soil test pH, used when no lime index is entered.

Expression

```
=IF(OR(LI=0, ISBLANK(LI)),  
    IF(AND(pH>=6.0, pH<=6.2), 1, IF(pH<6.0, 2, 0)),  
    IF(target_pH<=6.0, MAX(0, 54.2-0.78*LI),  
    IF(target_pH<=6.5, MAX(0, 65.5-0.94*LI),  
    IF(target_pH<=6.8, MAX(0, 71.2-1.02*LI), 0))))
```

New formula

Duplicate

Name

Lime Rec - MSU

Description

pH Targets: 6.0, 6.5, 6.8
BpH isn't supported yet. Define LI as numeric variable for now.

Output type

Product rate

Product

Lime

tons / ac

☐ Use interpolated nutrient layers ⓘ

NUMERIC VARIABLES ⓘ

#

LI

69

☐ Zone-based

☒ Prompt for value

#

target_pH

6.5

☐ Zone-based

☒ Prompt for value

+ Add

SOIL NUTRIENT VARIABLES ⓘ

pH

pH

pH Value

All

All

+ Add

CROP NUTRIENT COEFFICIENT VARIABLES ⓘ

+ Add

CROP VARIABLES ⓘ

+ Add

Generate

ASSIGNED GROWING SEASON CROP VARIABLES ⓘ

+ Add

Expression

```

-IF(OR(LI=0,ISBLANK(LI)),
  IF(AND(pH>=6.0,pH<=6.2),1,IF(pH<6.0,2,0)),
  IF(target_pH<=6.0,MAX(0,54.2-0.78*LI),
  IF(target_pH<=6.5,MAX(0,65.5-0.94*LI),
  IF(target_pH<=6.8,MAX(0,71.2-1.02*LI),0))))
          
```

Lime Rec - MSU in the formula editor

Sugar Beet Lime


Calculates tons of lime per acre from the lime index when it's below 69. Otherwise it uses soil pH: 1 ton when pH is 6.0 to 6.2, 2 tons below 6.0, and 0 above 6.2.

Output type Product rate
Product Lime, tons / ac

Variable	Type	Setting	What it does
pH	Soil nutrient	pH, pH Value	The soil test pH.
LI	Soil nutrient	Ton (T)	Lime index.

Expression

```
=IF(LI<69,
  65.5-(0.94*LI),
  IF(AND(pH>=6,pH<=6.2),
    1,
    IF(pH<6,
      2,
      0
    )
  )
)
```



Sugar Beet Lime in the formula editor

Sugar beet nitrogen

Two formulas calculate nitrogen from yield goal alone. The After Corn version adds 30 lb per acre.

Output type	Nutrient rate
Nutrient output	Nitrogen, Pounds per Acre
Description	MSU Nutrient

Variable	Type	Setting	What it does
YP	Numeric, Prompt for value on	Default 0	Yield goal. Enter it each time you run the formula, since the default is 0.


Sugar Beet Nitrate

```
=4*YP
```

Sugar Beet Nitrate_After Corn

```
=(4*YP)+30
```

The originals also include a Nitrogen soil variable. Neither expression uses it, so you can leave it out.



Sugar Beet Nitrate in the formula editor

Tri-State recommendations

Three of these formulas write a yield goal of 180 into the expression. To use a different yield goal, replace 180, or add a YP numeric variable with **Prompt for value** on and use YP in place of 180.

Tri-State K Rec

Potassium rate in steps by CEC and soil test potassium, plus maintenance.

Output type Nutrient rate
Nutrient output Potassium, Pounds per Acre

Variable	Type	Setting	What it does
K	Soil nutrient	Potassium, Parts Per Million (ppm)	The soil test potassium value.
CEC	Soil nutrient	Cation Exchange Capacity, mEq Per 100 Gram (mEq/100g)	Picks the potassium step to use.

Expression

```
=IF(AND(CEC<=5,K<=175),((175-K)*(1+(0.05*CEC)))+(180*0.27)+20),  
IF(AND(CEC<=10,K<=200),  
((200-K)*(1+(0.05*CEC)))+(180*0.27)+20,  
IF(AND(CEC<=20,K<=250),  
((250-K)*(1+(0.05*CEC)))+(180*0.27)+20,  
IF(AND(CEC<=30,K<=300),  
((300-K)*(1+(0.05*CEC)))+(180*0.27)+20,  
IF(AND(CEC>30,K<=350),  
((350-K)*(1+(0.05*CEC)))+(180*0.27)+20,  
IF(K<=400,65,0))))))
```

New formula

Duplicate

Name

Tri-State K Rec

Description

Output type

Nutrient rate

Nutrient output

Potassium

Pounds per Acre

Expression outputs a value in a different unit

Use interpolated nutrient layers

NUMERIC VARIABLES

Add

SOIL NUTRIENT VARIABLES

K

Potassium

Parts Per Million (ppm)

All

All

CEC

Cation Exchange Capacity

mEq Per 100 Gram (mEq/100g)

All

All

Add

CROP NUTRIENT COEFFICIENT VARIABLES

Add

CROP VARIABLES

Add

Generate

ASSIGNED GROWING SEASON CROP VARIABLES

Add

Expression

```

=IF (AND (CEC<=5, K<=175), ((175-K)*(1+(0.05*CEC)))+(180*0.27)+20,
IF (AND (CEC<=10, K<=200),
((200-K)*(1+(0.05*CEC)))+(180*0.27)+20,
IF (AND (CEC<=20, K<=250),
((250-K)*(1+(0.05*CEC)))+(180*0.27)+20,
IF (AND (CEC<=30, K<=300),
((300-K)*(1+(0.05*CEC)))+(180*0.27)+20,
IF (AND (CEC>30, K<=350),
((350-K)*(1+(0.05*CEC)))+(180*0.27)+20,
IF (K<=400, 65, 0))))))

```

Tri-State K Rec in the formula editor

Tri-State Nitrate Rec

Output type Nutrient rate
Nutrient output Nitrogen, Pounds per Acre

Variable	Type	Setting	What it does
N	Soil nutrient	Nitrogen, Parts Per Million (ppm)	The soil test nitrogen value.

Expression

```
=IF(AND(N<=30),(30-N)*2.5+(180*0.37),  
IF(AND(N<=60),(180*0.37),  
IF(AND(N<=70),(180*0.19),  
IF(AND(N<=79),30,0)))
```

Tri-State P Rec

Output type Nutrient rate
Nutrient output Phosphorus, Pounds per Acre

Variable	Type	Setting	What it does
P1	Soil nutrient	Phosphorus, Parts Per Million (ppm)	The soil test phosphorus value.

Expression

```
=IF(AND(P1<=30),(30-P1)*2.5+(180*0.37),  
IF(AND(P1<=60),(180*0.37),  
IF(AND(P1<=70),(180*0.19),  
IF(AND(P1<=79),30,0)))
```

New formula

Duplicate

Name

Tri-State P Rec

Description

Output type

Nutrient rate

Nutrient output

P Phosphorus P

Pounds per Acre

☐ Expression outputs a value in a different unit

☐ Use interpolated nutrient layers ⓘ

NUMERIC VARIABLES ⓘ

+ Add

SOIL NUTRIENT VARIABLES ⓘ

P1

P Phosphorus P

Parts Per Million (ppm)

All

All

+ Add

CROP NUTRIENT COEFFICIENT VARIABLES ⓘ

+ Add

CROP VARIABLES ⓘ

+ Add

Generate

ASSIGNED GROWING SEASON CROP VARIABLES ⓘ

+ Add

Expression

```
IF(AND(P1<=30),(30-P1)*2.5+(180*0.37),
IF(AND(P1<=60),(180*0.37),
IF(AND(P1<=70),(180*0.19),
IF(AND(P1<=79),30,0))))
```

Tri-State P Rec in the formula editor

Tristate Phosphorus Rec-IN,MI,OH

Returns a phosphorus rate for four yield goals: 150, 200, 250, and 300. Enter one of those four values. Any other value doesn't return a rate.

Output type	Nutrient rate
Nutrient output	Phosphorus, Pounds per Acre
Description	Yield Goal based Phos Rec

Variable	Type	Setting	What it does
YG	Numeric, Prompt for value on	Default 0	Yield goal: 150, 200, 250, or 300.

Expression

```
=IF(  
  YG=150, 55,  
  YG=200, 70,  
  YG=250, 90,  
  YG=300, 105  
)
```

The screenshot shows a formula editor window with the following fields and options:

- Name:** Tristate Phosphorus Rec-IN,MI,OH
- Description:** Yield Goal based Phos Rec
- Output type:** Nutrient rate
- Nutrient output:** P Phosphorus, Pounds per Acre
- ☐ Expression outputs a value in a different unit
- ☐ Use interpolated nutrient layers
- NUMERIC VARIABLES:** YG (value: 0), Zone-based (unchecked), Prompt for value (checked)
- SOIL NUTRIENT VARIABLES:** Add
- CROP NUTRIENT COEFFICIENT VARIABLES:** Add
- CROP VARIABLES:** Add, Generate
- ASSIGNED GROWING SEASON CROP VARIABLES:** Add

Tristate Phosphorus Rec-IN,MI,OH in the formula editor